



32bits/16bits 无符号除法运算

```

;*****
;*****
;程序名称: _32DIV16
;程序功能: 32bits 除以 16bits 无符号除法运算子程序
;入口参数: ACCBH/ACCBL/ACCAH/ACCAL 存放被除数
;          H_IER/L_IER 中存放除数
;出口参数: ACCDH/ACCDL/ACCCH/ACCCL 存放商
;          ACRTH/ACCEL 存放余数
;*****

```

_32DIV16:

```

    CLR        ACRTH
    CLR        ACRTL
    LDIA       .32
    LD         ACCNT, A

```

_32DIV16_1:

```

    CLR        STATUS, C
    RLCR       ACCAL
    RLCR       ACCAH
    RLCR       ACCBL
    RLCR       ACCBH
    RLCR       ACRTL
    RLCR       ACRTH
    SZB        STATUS, C
    JP         _32DIV16_3
    LD         A, H_IER
    SUBA       ACRTH

```



```
SNZB      STATUS, C
JP        _32DIV16_4
SNZB      STATUS, Z
JP        _32DIV16_3
_32DIV16_2:
LD        A, L_IER
SUBA      ACRTL
SNZB      STATUS, C
JP        _32DIV16_4
_32DIV16_3:
LD        A, L_IER
SUBR      ACRTL
SNZB      STATUS, C
DECR      ACRTH
LD        A, H_IER
SUBR      ACRTH
SETB      STATUS, C
_32DIV16_4:
RLCR      ACCCL
RLCR      ACCCH
RLCR      ACCDL
RLCR      ACCDH
SZDECR    ACCNT
JP        _32DIV16_1
RET       00H
```



;应用实例

;使用前为子程序声明缓存变量(均为局部变量,可与其它局部变量复用)

ACCBH	EQU	?	;被除数 24-31 位
ACCBL	EQU	?	;被除数 16-23 位
ACCAH	EQU	?	;被除数 8-15 位
ACCAL	EQU	?	;被除数 0-7 位
H_IER	EQU	?	;除数 8-15 位
L_IER	EQU	?	;除数 0-7 位
ACCDH	EQU	?	;商 24-31 位
ACCDL	EQU	?	;商 16-23 位
ACCCH	EQU	?	;商 8-15 位
ACCCL	EQU	?	;商 0-7 位
ACRTH	EQU	?	;余数 8-15 位
ACRTL	EQU	?	;余数 0-7 位
ACCNT	EQU	?	;移位运算次数

;以下声明用户变量

USERRAM1	EQU	?
USERRAM2	EQU	?
USERRAM3	EQU	?
USERRAM4	EQU	?

;应用程序片段



;上接其它程序

SHOW_PROGRAM:

```
LDIA      080H
LD        ACCBH, A
LD        ACCBL, A
LDIA      050H
LD        ACCAH, A
LD        ACCAL, A      ;赋值给被除数
LDIA      060H
LD        H_IER, A
LD        L_IER, A      ;赋值给除数
CALL      _32DIV16      ;调用 32/16 子程序
LD        A, ACCDH
LD        USERRAM1, A
LD        A, ACCDL
LD        USERRAM2, A
LD        A, ACCCH
LD        USERRAM3, A
LD        A, ACCCL
LD        USERRAM4, A   ;将商保存到用户寄存器
```

;下接其它程序
